*==========Why_Window_With_Just_A_Cosine?==========

Windowing with just a simple 100% amplitude modulated cosine can do much in terms of handling Spectrum leakage. The following shows the leakage for a 14.1KHz signal before and after applying the hanning window.

==========





So why stop there? Does it need to be just 100% AM like Hamming? How about adding a little 2nd harmonic in the case of Blackman? How about adding a little 3nd harmonic, etc..

==========

```
*V_SIN#    NODE_P  NODE_N DC     VALUE  SIN(   V_DC   AC_MAG FREQ   DELAY  FDamp)
Vsig       Vsig    0       DC     0      SIN(   0      1      14.1k                 )
VCos1      VCos1   0       DC     0      SIN(   0      1      1k      -.25m         )
VCos2      VCos2   0       DC     0      SIN(   0      1      2k      -.125m        )
VCos3      VCos3   0       DC     0      SIN(   0      1      3k      -.0833333m    )

Brect      OUTR    0       V =    V(Vsig)
Bhann      OUTHN   0       V =    V(Vsig)*(.5      -.5000000*v(VCos1))
Bhamming   OUTHM   0       V =    V(Vsig)*(.54     -.4600000*v(VCos1))
Bblack     OUTBK   0       V =    V(Vsig)*(.42     -.5000000*v(VCos1)  +.080000*v(VCos2))
BblackHar  OUTBH   0       V =    V(Vsig)*(.35875  -.4882900*v(VCos1)  +.141280*v(VCos2) -.0116800*v(VCos3))
BblackNut  OUTBN   0       V =    V(Vsig)*(.3635819-.4891775*v(VCos1)  +.136595*v(VCos2) -.0106411*v(VCos3))
```


dB
dB(mag(v(outhm)))
0.0
-20.0
-40.0
-60.0
-80.0
-100.0
-120.0
-140.0
decibel
10^3
10^4
10^5
frequency Hz


dB
dB(mag(v(outbk)))
0.0
-20.0
-40.0
-60.0
-80.0
-100.0
-120.0
-140.0
decibel
10^3
10^4
10^5
frequency Hz


dB
dB(mag(v(outbh)))
0.0
-20.0
-40.0
-60.0
-80.0
-100.0
-120.0
-140.0
decibel
10^3
10^4
10^5
frequency Hz



==========Full_Netlist_For_Copy_Paste==========

```
Tweeking_Hann
.Option srcsteps = 1 set Gmin = 1.0000E-02
*=======Circuit_Netlist=========================
*V_SIN#    NODE_P  NODE_N DC     VALUE  SIN(   V_DC   AC_MAG FREQ   DELAY  FDamp)
Vsig       Vsig    0      DC     0      SIN(   0      1      14.1k                )
Vtri       Vtri    0      DC     0      PWL(   0      0      .5m      1     1m    0 )
VCos1      VCos1   0      DC     0      SIN(   0      1      1k      -.25m         )
VCos2      VCos2   0      DC     0      SIN(   0      1      2k      -.125m        )
VCos3      VCos3   0      DC     0      SIN(   0      1      3k      -.0833333m    )
Brect      OUTR    0      V =    V(Vsig)
Bhann      OUTHN   0      V =    V(Vsig)*(.5      -.5000000*v(VCos1))
Bhamming   OUTHM   0      V =    V(Vsig)*(.54     -.4600000*v(VCos1))
Bblack     OUTBK   0      V =    V(Vsig)*(.42     -.5000000*v(VCos1)  +.080000*v(VCos2))
BblackHar  OUTBH   0      V =    V(Vsig)*(.35875  -.4882900*v(VCos1)  +.141280*v(VCos2) -.0116800*v(VCos3))
BblackNut  OUTBN   0      V =    V(Vsig)*(.3635819-.4891775*v(VCos1)  +.136595*v(VCos2) -.0106411*v(VCos3))

.control
set        pensize = 2
*TRAN      TSTEP TSTOP TSTART TMAX ?UIC?

tran       .1u       1m    0      .1u
set        specwindow=   "rectangular"
spec       1k        100k   1k     v(outr)
plot       dB(mag(v(OUTr)))    xlog ylimit -140 0

tran       .1u       1m    0      .1u
set        specwindow=   "rectangular"
spec       1k        100k   1k     v(outhn)
plot       dB(mag(v(outhn)))    xlog ylimit -140 0

tran       .1u       1m    0      .1u
set        specwindow=   "rectangular"
spec       1k        100k   1k     v(outhm)
plot       dB(mag(v(outhm)))    xlog ylimit -140 0

tran       .1u       1m    0      .1u
set        specwindow=   "rectangular"
spec       1k        100k   1k     v(outbk)
plot       dB(mag(v(outbk)))    xlog ylimit -140 0

tran       .1u       1m    0      .1u
set        specwindow=   "rectangular"
spec       1k        100k   1k     v(outbh)
plot       dB(mag(v(outbh)))    xlog ylimit -140 0

tran       .1u       1m    0      .1u
set        specwindow=   "rectangular"
spec       1k        100k   1k     v(outbn)
plot       dB(mag(v(outbn)))    xlog ylimit -140 0

.endc
.end


7.29.10_12.02PM
dsauersanjose@aol.com
Don Sauer
http://www.idea2ic.com/
```